


Extremwertaufgabe: Silberschmied Spitzpokal

Prof. Dr. Dörte Haftendorn, MuPAD 4, Sept.06 Update

Web: <http://haftendorn.uni-lueneburg.de> www.mathematik-verstehen.de

Mathix ist Silberschmied geworden. Er möchte Pokale schmieden, bei denen auf einen Kegel mit Radius r und Höhe h ein offener Zylinder aufgesetzt ist. Er hat versucht auszurechnen, welche Zylinderhöhe und welchen Radius er bei festem Volumen (z.B. 2 Liter) nehmen muss, damit er minimalen Silberverbrauch hat. Mathilde sagt: "Am besten lässt du den Zylinder weg, das war bei dem Pokal mit der Halbkugel auch am besten." Hat Mathilde Recht? (Variante mit halbkugelförmigem Unterteil siehe Extradatei)



[animieren durch Anklicken!](#)

Zielgröße: Silberverbrauch, entspricht der Oberfläche, setzt sich zusammen aus dem Zylindermantel und dem Kegelmantel ($hh=h$, weil h später gebraucht wird, s =Mantellinie):

$$S_{\text{formel}} := 2 \cdot \pi \cdot r \cdot hh + \pi \cdot r \cdot s;$$

$$2 \cdot \pi \cdot hh \cdot r + \pi \cdot r \cdot s$$

Nebenbedingung, Volumen 2 Liter = 2000 cm^3 , setzt sich zusammen aus Kegelvolumen und Zylindervolumen

delete V:

$$s := \sqrt{2} \cdot r;$$

$$\text{NB} := V = \pi \cdot r^2 \cdot hh + \frac{1}{3} \cdot \pi \cdot r^3;$$

$$\sqrt{2} \cdot r$$

$$V = \frac{\pi \cdot r^3}{3} + \pi \cdot hh \cdot r^2$$

Nach r ließe sich diese Gleichung gar nicht leicht auflösen, daher Auflösen nach hh

$$\text{assume}(r > 0) : \text{solve}(\text{NB}, hh);$$

$$\left\{ \frac{V - \frac{\pi \cdot r^3}{3}}{\pi \cdot r^2} \right\}$$

$$\text{expand}(\text{op}(\%))$$

$$\frac{V}{\pi \cdot r^2} - \frac{r}{3}$$

`h:=r->V/(PI*r^2)-1/3*r;h(r);`

$$r \rightarrow \frac{V}{\pi \cdot r^2} - \frac{r}{3}$$

$$\frac{V}{\pi \cdot r^2} - \frac{r}{3}$$

Aufstellen der Zielfunktion S in Abhängigkeit von r allein

`S:=r->2*PI*r*h(r)+PI*r*s;S(r); expand(S(r)):`

$$r \rightarrow 2 \cdot \pi \cdot r \cdot h(r) + \pi \cdot r \cdot s$$

$$\pi \cdot \sqrt{2} \cdot r^2 - 2 \cdot \pi \cdot r \cdot \left(\frac{r}{3} - \frac{V}{\pi \cdot r^2} \right)$$

Man kann die Ausgabe markieren und in eine neue Ausgabezelle ziehen.

`expand(2^(1/2)*PI*r^2 - 2*PI*r*(1/3*r - 1/PI*V/r^2));`

$$\frac{2 \cdot V}{r} - \frac{2 \cdot \pi \cdot r^2}{3} + \pi \cdot \sqrt{2} \cdot r^2$$

`S:=r->2*V/r - 2/3*PI*r^2 + 2^(1/2)*PI*r^2: S(r); //Zielfunktion`


$$\frac{2 \cdot V}{r} - \frac{2 \cdot \pi \cdot r^2}{3} + \pi \cdot \sqrt{2} \cdot r^2$$

Aufbau der Zielfunktion aus Bausteinen:

Nun wird das Volumen auf 2 Liter=2000cm³ festgesetzt.

`V:=2000: //Längen in cm`

`plotfunc2d(2*V/r,S(r), - 2/3*PI*r^2 + 2^(1/2)*PI*r^2,r=0..10, ViewingBoxYRange=0..1000);`



Als Summe aus Parabel und Hyperbel hat die Zielfunktion ein gesichertes Minimum, etwa bei $r=10$.

Berechnung:

$S'(r);$

$\text{solve}(S'(r)=0, r);$

$$2 \cdot \pi \cdot \sqrt{2} \cdot r - \frac{4 \cdot \pi \cdot r}{3} - \frac{4000}{r^2}$$

$$\left\{ \sqrt[3]{-\frac{6000}{2 \cdot \pi - 3 \cdot \pi \cdot \sqrt{2}}} \right\}$$

$rs := \text{float}(\%[1])$

9.478666144

Welches ist das maximale r , für welches die 2 Liter ganz in dem Kegel sind?

Wenn r noch größer würde, wäre der Pokal ja zu groß.

Berechnung aus dem Kegelvolumen:

$V:=2000:$

$\text{solve}(V=1/3*PI*r^3, r);$

$rm := (\text{op}(\%)) ; rmf := \text{float}(rm);$

$$\left\{ \sqrt[3]{\frac{6000}{\pi}} \right\}$$

$$\sqrt[3]{\frac{6000}{\pi}}$$

12.40700982

Da ergibt sich ein größerer Wert, darum hat der Pokal mit minimalem Verbrauch einen zylindrischen Rand mit folgender Höhe:

```
simplify(h(rs));
```

```
3.92619207
```

Damit ist klar, dass Mathilde nicht Recht hat, der Pokal für 2 Liter mit minimalem Silberverbrauch hat einen Zylinderrand von fast 4 cm. Die Silberoberfläche ist dann

```
simplify(S(rs));
```

```
float(%)
```

```
633.0004569
```

```
633.0004569
```

#####

Herstellung der Graphen:


Sinnvolle optische Begrenzung für h könnte 10 sein. Wie groß ist dann r

```
solve(h(r)=10,r)
```

```
(0, ∞) ∩ RootOf( $\pi \cdot X1^3 + 30 \cdot \pi \cdot X1^2 - 6000$ , X1)
```

Exakt geht es nicht, also Ablesen:

```
plotfunc2d(h(r),10,r=-10..10,ViewingBoxYRange=0..12)
```



Numerische Bestimmung:

```
lo:=numeric::solve(h(r)=10,r);
```

```
lm:=numeric::solve(h(r)=0,r)
```

```
{-27.46883186, -9.699440553, 7.16827241}
```

```
{12.40700982, -6.203504909 - 10.74478569 · i, -6.203504909 + 10.74478569
```

```
rmin:=lo[3];
```

```
rmax:=lm[1];
```

```
7.16827241
```

```
12.40700982
```


```
gerli2d:=plot::Line2d([-r,0],[0,-r],r=rmin..rmax):
```

```
gerre2d:=plot::Line2d([r,0],[0,-r],r=rmin..rmax):
```

```

zy2d1:=plot::Line2d([-r,0],[-r,h(r)],r=rmin..rmax ):
zy2d2:=plot::Line2d([r,0],[r,h(r)],r=rmin..rmax):
plot(gerli2d, gerre2d,zy2d1,zy2d2, ViewingBoxYRange=-rmax..rmax,
      LineWidth=1, Scaling=Const

```




animieren durch Anklicken!

```

hkeg3d:=plot::Surface([k*cos(t),k*sin(t),-r+k],
                      t=0..2*PI,k=0..r , r=rmin..rmax):
zyl3d:=plot::Surface([r*cos(t),r*sin(t),hh],t=0..2*PI,hh=0..r,
                    r=rmin..rmax, ViewingBoxZRange=-rmax..rmax,
                    AnimationStyle=BackAndForth):
plot(hkeg3d,zyl3d)

```



animieren durch Anklicken!

Gezeigt sind verschiedene Pokale mit 2 Liter Inhalt.

#####


Darstellung der Zielfunktion allein und eines animierten Silververbrauchspunktes

```


Sg:=plot::Function2d(S(r),r=rmin..rmax, LineColor=[1,0,0]):

```


2-Liter-Pokal



Silberverbrauch



 animieren durch Anklicken!